
Abstract: This paper deals with the study and analysis of several rational approximations to approach
the behavior of arbitrary-order differentiators and integrators in the frequency domain. From the
Riemann–Liouville, Grünwald–Letnikov and Caputo basic definitions of arbitrary-order calculus
until the reviewed approximation methods, each of them is coded in a Maple 18 environment
and their behaviors are compared. For each approximation method, an application example is
explained in detail. The advantages and disadvantages of each approximation method are discussed.
Afterwards, two model order reduction methods are applied to each rational approximation and
assist a posteriori during the synthesis process using analog electronic design or reconfigurable
hardware. Examples for each reduction method are discussed, showing the drawbacks and benefits.
To wrap up, this survey is very useful for beginners to get started quickly and learn arbitrary-order
calculus and then to select and tune the best approximation method for a specific application in the
frequency domain. Once the approximation method is selected and the rational transfer function is
generated, the order can be reduced by applying a model order reduction method, with the target of
facilitating the electronic synthesis.

Keywords: arbitrary-order calculus; Oustaloup’s approximation; refined Oustaloup’s approximation;
Matsuda’s approximation; continued fraction expansion approximation; Charef’s approximation;
Carlson’s approximation; curve fitting approximation; modified stability boundary locus approxima-
tion; model order reduction

1. Introduction

Arbitrary-order calculus is an important issue that has attracted the attention of re-
search institutes, academic associations, industrial areas and foundry companies. It has
shown a better way of modeling and controlling many financial, biological, chemical,
physical, electrical and control phenomena [1–3]. Capacitors and inductors are dissipative
two-terminal passive elements widely used to describe integer-order differentiators, in-
tegrators and state-space systems. However, to date, there is no robust passive element
with arbitrary characteristics so that it can be used to describe the behavior of a system of
arbitrary-order differential equations that models natural phenomena or human-made sys-
tems more realistically. In response to this problem, integer-order rational approximations
in the frequency domain were developed to approximate the arbitrary order with low level
of error and wide bandwidth, such as Oustaloup’s [4], refined Oustaloup’s [5], Charef’s [6],

Carlson’s [7,8], Matsuda’s [9], etc. Examples of its use in several applications can be found
in the literature. For instance, a synthesis methodology of arbitrary-order chaotic systems
was discussed in [10]. The analysis and analog design of arbitrary-order charge/flux
controlled memristor emulators of incremental/decremental type was described in [11].
In [3,12,13] the analog design of arbitrary-order proportional-integral-derivative controllers
was reported. The arbitrary-order lead/lag compensator was investigated in [14,15]. The
design of double exponent arbitrary-order filters and power law filters were investigated
in [16,17], respectively, and so on. Note that in each application, a different arbitrary-order
approximation method was used; once the rational approximation is deduced, this can be
synthesized using analog or digital platforms. Therefore, this survey not only describes and
analyzes the main approximation methods reported to date, but two model order reduction
methods are also reviewed in order to obtain rational transfer functions of reduced order
and with the target of simplifying the electronic synthesis process. Unlike [18,19] and to
help beginners get started quickly and learn arbitrary-order calculus, the corresponding
Maple 18 code of each approximation method is herein reported.

The paper is organized as follows. Section 2 deals with the three main basic definitions
on arbitrary-order calculus. The Maple code for each of them is also introduced in this
section. Section 3 describes nine rational approximation methods in the frequency domain.
Further, the Maple 18 code for each approximation method is not only presented, but an ex-
ample of application by each method is also developed. The advantages and disadvantages
of each method are also discussed. Section 4 deals with the analysis of two model order
reduction methods. Examples for each method are also discussed in this section, showing
the drawbacks and benefits. Finally, the results and conclusions are given in Section 5.

2. Basic Definitions
2.1. Arbitrary-Order Integral and Derivative of Riemann–Liouville

The arbitrary-order integral in the sense of Riemann–Liouville is defined as [1,3]

a Jα
t f (t) =

1
Γ(α)

∫ t

a

f (τ)
(t− τ)1−α

dτ (1)

where a and t are the lower and upper limits of integration, Γ(·) is the gamma function,
α ∈ <+ is the arbitrary order and f (τ) describes a causal function of time. This first basic
definition can easily be code in Maple 18 environment, as described in Table 1.

Table 1. Maple 18 code of (1).

restart:
assume(alpha <= 1); additionally(0 < alpha);
AOI_RL := proc(alpha,f)
1/GAMMA(alpha)*int(f*(t-tau)∧(alpha-1),tau = 0..t,‘AllSolutions’) assuming t > 0;
end proc:

Let us consider an example on the use of Table 1. Defining f := τ2 − 2τ + 1 and
α := 0.65 we compute the arbitrary-order integral as

S1 := AOI_RL(alpha, f );

S1 := 0.5081609661t(53/20) − 1.346626560t(33/20) + 1.110966912t(13/20)
(2)

making a variable change and defining α = 0.35 we have

alpha := 0.35 : t := tau : Sa := S1; unassign(′t′);

Sa := 0.5081609661τ(53/20) − 1.346626560τ(33/20) + 1.110966912τ(13/20)

S2 := AOI_RL(alpha, Sa);

S2 := 0.3333333334t3 − t2 + t

(3)

the last equation in (3) is the first-order integral of the original function f (τ) described
above and was found by applying (1) twice. Despite the ease of application, the main
disadvantage of this basic definition is that the analytical solution of complex functions
cannot be found easily and the solution is often a function of other nested functions.
Moreover, the inverse process of (1), i.e., the arbitrary-order derivative, is defined as

Dα f (t) =
dp

dtp

[
1

Γ(p− α)

∫ t

a

f (τ)
(t− τ)1+α−p dτ

]
(4)

where p is a positive integer such that p− 1 < α ≤ p. Based on the previous programming
code, (4) is coded in Table 2.

Table 2. Maple 18 code of (4).

restart:
assume(p-1 < alpha);additionally(alpha <= p);
AOD_RL := proc(alpha,f,p)
diff(1/GAMMA(p-alpha)*int((t-tau)∧(p-alpha-1)*f,tau = 0..t),t$p) assuming t >= 0;
end proc:

Let us consider an example on the use of Table 2. Defining f := τ3

3 − τ2 + τ, p := 1
and α := 0.15 we get

S1 := AOD_RL(alpha, f , p);

S1 := 0.4011444710t(57/20) − 1.143261742t(37/20) + 1.057517112t(17/20)
(5)

making, again, a variable change and updating α = 0.85, we have

alpha := 0.85 : t := tau : Sa := S1; unassign(‘t’);

Sa := 0.4011444710τ(57/20) − 1.143261742τ(37/20) + 1.057517112τ(17/20)

S2 := AOD_RL(alpha, Sa, p);

S2 := 0.99999t2 − 2t + 1

(6)

the last equation in (6) is the first-order derivative of f (τ) mentioned before and going
back to the original function used in the first example. Note that (4) was also applied twice.
Aside from the aforementioned disadvantage of limiting to simple functions, another
disadvantage of (4) is that the initial conditions of the test function cannot be considered.
This is a serious problem from a practical engineering standpoint.

2.2. Arbitrary-Order Integral and Derivative of Grünwald–Letnikov

The arbitrary-order integral in the sense of Grünwald–Letnikov is defined as [1,3]

a Jα
t f (t) = lim

h→0
hα

t−a
h

∑
p=0

Γ(α + p)
p!Γ(α)

f (t− ph) (7)

where h is the integration step. This third basic definition is coded in Maple 18 environment
as illustrated in Table 3. Defining α = 0.65, h = 0.01, tf = 3, a = 0 and f := t2 − 2t + 1, we
got the numerical behavior of arbitrary-order integral of the function as

alpha := 0.65 : h := 0.01; t f := 3 : a := 0 :

f := subs(t = t− p ∗ h, f ) : f := unapply( f , t, p, h) :

S1 := AOI_GL(alpha, h, t f , a) :

(8)

the second line of (8) defines a variable change in f. Because (7) is a discrete function, the
time-domain response of complex functions can numerically be computed. However, a
level of error is glimpsed compared with its continuous part. This error can be reduced,
making h→ 0, but the CPU time increases. Therefore, there is a trade-off between accuracy
and speed. Another disadvantage is related with the memory length since (7) is an iterative
procedure and the sum in its scheme becomes longer and longer. Note that whereas the
analytical solution is given at the second line of (2), its numerical response is shown in
Figure 1 (blue-line).

Table 3. Maple 18 code of (7).

restart:
f := subs(t = t − p*h,f): f := unapply(f,t,p,h):
AOI_GL := proc(alpha,h,tf,a)
local t,p,Sa,S1 := []:
for t from 0 by h to tf do

Sa := 0:
for p from 0 by 1 to (t-a)/h do

Sa := Sa + h∧alpha*GAMMA(alpha + p)/(p!*GAMMA(alpha))*f(t,p,h):
od:
S1 := [op(S1),[t,Sa]]:

od:return(S1):
end proc:

Figure 1. Time-domain response of (8) (blue line) and (10) (red line).

Moreover, the arbitrary-order derivative is defined as

Dα f (t) = lim
h→0

1
hα

t−a
h

∑
p=0

(−1)p Γ(α + 1)
p!Γ(1 + α− p)

f (t− ph) (9)

By virtue of the form of (9) and according to Table 3, the Maple 18 code is given
in Table 4. Defining α = 0.65, h = 0.01, tf = 3, a = 0 and f := 0.5081609661 ∗ t(53/20) −

1.346626560 ∗ t(33/20)+ 1.110966912 ∗ t(13/20), we obtain the numerical behavior of arbitrary-
order derivative of the function as

alpha := 0.65 : h := 0.01; t f := 3 : a := 0 :

f := subs(t = t− p ∗ h, f ) : f := unapply( f , t, p, h) :

S1 := AOD_GL(alpha, h, t f , a) :

(10)

The time-domain numerical response is shown in Figure 1 (red-line). Apart from
the disadvantages mentioned above, another drawback of (9) is related with the initial
conditions of f, which cannot be included.

Table 4. Maple 18 code of (9).

restart:
f := subs(t = t − p*h,f): f := unapply(f,t,p,h):
AOD_GL := proc(alpha,h,tf,a)
local t,p,Sa,S1 := []:
for t from 0 by h to tf do

Sa := 0:
for p from 0 by 1 to (t-a)/h do

Sa := Sa + (−1)∧p/h∧alpha*GAMMA(alpha + 1)/(p!*GAMMA(1 + alpha-p))*f(t,p,h):
od:
S1 := [op(S1),[t,Sa]]:

od:return(S1):
end proc:

2.3. Arbitrary-Order Derivative of Caputo

The arbitrary-order derivative in the sense of Caputo is defined as [1,3]

Dα f (t) =
1

Γ(p− α)

∫ t

0

f (p)(τ)

(t− τ)1+α−p dτ (11)

This last basic definition is easily coded in the Maple 18 environment, as given in
Table 5.

Table 5. Maple 18 code of (11).

restart:
assume(p − 1 < alpha);additionally(alpha <= p);
AOD_C := proc(alpha,f,p)
1/GAMMA(p-alpha)*int(diff(f,tau$p)/(t-tau)∧(1 + alpha-p),tau = 0..t) assuming t >= 0;
end proc:

Let us consider an example on the use of Table 5. Defining f := τ3

3 − τ2 + τ, p := 1
and α := 0.15 we compute the arbitrary-order derivative as

S1 := AOD_C(alpha, f , p);

S1 := 0.4011444709t(57/20) − 1.143261742t(37/20) + 1.057517111t(17/20)
(12)

making, again, a variable change and updating α = 0.85, we have

alpha := 0.85 : t := tau : Sa := S1; unassign(′t′);

Sa := 0.4011444709τ(57/20) − 1.143261742τ(37/20) + 1.057517111τ(17/20)

S2 := AOD_C(alpha, Sa, p);

S2 := 0.99999999t2 − 2t + 0.99999997

(13)

The last equation of (13) is the first-order derivative of f (τ) mentioned before, where
(11) was also applied twice. A main advantage of this basic definition is that the initial
conditions of the test function can be considered, i.e., f (0), f ′(0),. . . , f p−1(0). However, the
analytical solution of complex functions cannot be easily computed, and the solution is
again a function of other nested functions.

Moreover, natural phenomena are often modeled by state-space representation, which
is a system of integer- or arbitrary-order differential equations [2]. Once the physical phe-
nomena are modeled, the following task is to find their solution. From an engineering point
of view, there are two ways to solve the state-space system: frequency-domain methods [18]
and time-domain methods [20]. Note that the main basic definitions of arbitrary-order
calculus are of the second type. Other reported methods in the time domain are the
Adomian decomposition method [21], predictor–corrector approach based on one-step
Adams–Bashforth–Moulton method [22–25] and Chebyshev collocation method [26,27].
However, the main disadvantage of all these methods is that numerical simulations can only
be performed, and some of them are not robust enough to be encoded on reconfigurable
hardware. In this context, frequency-domain methods are based on the approximation of
the transition function 1

sα through pole-zero pairs. Although this type of approximation
has two main drawbacks—(1) the approximations of 1

sα is valid only for few values of α
and (2) the difference between the original behavior and approximate is still unknown—its
major advantage is that integer-order rational polynomials can be synthesized with discrete
analog circuits or encoded in reconfigurable hardware and, as a consequence, experimental
results can be observed.

3. Rational Approximations in the Frequency Domain

In the analog domain, the expression Z(s) = k0
sα , where k0 is a constant called fractance

device. Several frequency-domain approximation methods have been reported in the
literature that can be used to compute the integer-order rational approximations of the
fractance, as described below.

3.1. Oustaloup’s Approximation

This approximation method is based on a recursive distribution of zeros and poles into
a frequency interval (wl , wh), where wl and wh are the low and high transitional frequencies
expressed in rad/s [4]. Thus, the target is to approximate the function of the form

H(s) = sα, α ∈ <+ (14)

by an integer-order rational approximation

sα = wα
h

N

∏
k=1

s + zk
s + pk

, 0 < α < 1 (15)

The zeros are computed as

zk = wl

(
wh
wl

) 2k−1−α
2N

(16)

and the poles are computed as

pk = wl

(
wh
wl

) 2k−1+α
2N

(17)

where N is the approximation order. Equations (15)–(17) are coded in Maple 18 as given in
Table 6.

Table 6. Oustaloup’s approximation coded in Maple 18 environment.

restart:Digits := 6:
Oustaloup := proc(alpha,wl,wh,N)
local p,z,Ha;
z := wl*(wh/wl)∧((2*k − 1 − alpha)/(2*N)):
p := wl*(wh/wl)∧((2*k − 1 + alpha)/(2*N)):
Ha := wh∧alpha*factor(product((s + z(k))/(s + p(k)),k = 1..N)):
return(Ha):
end proc:

Let us consider an example on the use of Table 6. Defining α = 0.5, wl = 0.01, wh = 100
and N = 4, the rational approximation is computed as

alpha := 0.5 : wl := 0.01 : wh := 100 : N := 4 :

Ha := Oustaloup(alpha, wl, wh, N);

Ha := 10.0
(s + 0.0177828)(s + 0.177828)(s + 1.77828)(s + 17.7828)
(s + 0.0562341)(s + 0.562341)(s + 5.62341)(s + 56.2341)

Ha := expand(numer(Ha))/expand(denom(Ha));

Ha :=
10s4 + 197.567s3 + 354.523s2 + 62.4761s + 1

s4 + 62.4761s3 + 354.522s2 + 197.566s + 9.99994

(18)

the third line of (18) expresses the rational approximation as a product of poles and zeros,
and the fifth line as a quotient of fourth-order rational polynomials. Note that here sα

was approximated by (18), that is, the arbitrary-order derivative. To obtain 1/sα, i.e., the
arbitrary-order integral, the numerator and denominator of (18) must be interchanged.
In this sense, whereas the magnitude and phase responses of the arbitrary-order integral
approximation along with the ideal behavior (black line) are shown in Figure 2a,c, the errors
for each case are depicted in Figure 2b,d, respectively. All magnitude and phase waveforms
along with their errors are colored with a blue line. Advantages of this approximation
method are that (1) it allows to obtain rational polynomials in a wide range of frequencies
and (2) for the same iteration number, the rational approximation has a lower order
compared with other approximation methods. However, its major disadvantage is that the
phase error does not decrease in the frequency interval despite increasing the number of
iterations. As a consequence, high-order rational approximations are generated and later,
this becomes a more complex and tedious task when the rational polynomial is synthesized
with discrete analog circuits or encoded with reconfigurable hardware.

(a) (b)

Figure 2. Cont.

(c) (d)
Figure 2. Ideal behavior (black line), Oustaloup’s (blue line), refined Oustaloup’s (red line), Charef’s
V1 (brown line), ideal behavior (gray line) of (26), Charef’s V2 (maroon line) and Carlson’s (gold
line) approximation for 1

s0.5 . (a) Magnitude response. (b) Magnitude error. (c) Phase response.
(d) Phase error.

3.2. Refined Oustaloup’s Approximation

This approximation method is defined as [5]

sα =

(
d
b

wh

)α ds2 + bwhs
d(1− α)s2 + bwhs + dα

N

∏
k=1

s + zk
s + pk

, 0 < α < 1 (19)

where b = 10 and d = 9 are adjustment parameters and their values were found by exper-
imentation. The zeros and poles are computed by using (16) and (17). Note that in [5],
a slight modification was reported compared with (19), where the bottom index of the
product starts from -N instead of 1, as in the original Outstaloup’s notation. Nevertheless,
both approximations, reported by Equation (23) in [5] and (19) are equivalent. According
to Table 6, (19) is easily coded in the Maple 18 environment as shown in Table 7.

Table 7. Refined Oustaloup’s approximation coded in Maple 18 environment.

restart:Digits := 6:
Ref_Oustaloup := proc(alpha,wl,wh,N)
local p,z,Ha,b := 10,d := 9;
z := wl*(wh/wl)∧((2*k − 1 − alpha)/(2*N)):
p := wl*(wh/wl)∧((2*k − 1 + alpha)/(2*N)):
Ha := factor((d*wh/b)∧alpha*(d*s∧2 + b*wh*s)/(d*(1 − alpha)*s∧2 + b*wh*s + d*alpha))

*factor(product((s + z(k))/(s + p(k)),k = 1..N)):
return(Ha):
end proc:

Using α = 0.5, wl = 0.01, wh = 100 and N = 4, the rational approximation is computed as

alpha := 0.5 : wl := 0.01 : wh := 100 : N := 4 :

Ha := Re f _Oustaloup(alpha, wl, wh, N);

Ha :=
18.9737s(s + 111.111)(s + 0.0177)(s + 0.17782)(s + 1.77828)(s + 17.7828)
(s + 222.21)(s + 0.0045)(s + 0.05623)(s + 0.5623)(s + 5.6234)(s + 56.234)

Ha := expand(numer(Ha))/expand(denom(Ha));

Ha :=
18.9737s6 + 2483.05s5 + 42323.5s4 + 74859.0s3 + 13173.1s2 + 210.819s

s6 + 284.698s5 + 14239.1s4 + 79043.0s3 + 44268.6s2 + 2419.79s + 9.99994

(20)

The magnitude and phase responses of the arbitrary-order integral approximation
are illustrated in Figure 2a,c while their errors are shown in Figure 2b,d. All waveforms
are colored with a red line. Advantages of this approximation method are that (1) it has a
better fitting quality around the frequency band boundaries in both magnitude and phase
responses, as can be seen in Figure 2 (red line) and (2) for the same number of iterations,
the error decreases compared with the Oustaloup’s method. The main disadvantage found
is that the order of the rational approximation increases, as one can see in (20), generating
all the drawbacks mentioned above.

3.3. Charef’s Approximation Version 1

The Charef’s approximation method was reported in [6] to approximate the transfer
function as a function of singularities of poles an zeros into a frequency bandwidth (wl , wh),
which depends on an approximation error ε (dB) previously defined. Thus, the objective is
to approximate the function

H(s) =
1
sα
≈ ∏N−1

k=0

(
1 + s

z
)

∏N
k=0

(
1 + s

p

) , 0 < α < 1 (21)

where the coefficients are computed for obtaining a maximum deviation according to
ε. Thus

a = 10
ε

10(1−α) , b = 10
ε

10α , (22)

therefore, the distribution of the poles and zeros becomes

p0 = wl10
ε

20α , p = (ab)k p0, z = a(ab)k p0 (23)

and the total number of singularities is determined as

N = integer

 log
(

wh
p0

)
log(ab)

+ 1 (24)

Here, integer indicates the integer part of the mathematical operation. Equations (21)–(24)
are coded in Maple 18 as shown in Table 8.

Table 8. Code in Maple 18 environment of (21).

restart:Digits := 6:
Charef_V1 := proc(alpha,wl,wh,epsilon)
local p0,p,z,a,b,k,Ha,N;
p0 := wl*10∧(epsilon/(20*alpha)):
a := 10∧(epsilon/(10*(1-alpha))):
b := 10∧(epsilon/(10*alpha)):
N := ceil(log10(wh/p0)/log10(a*b)):
p := (a*b)∧k*p0: z := (a*b)∧k*a*p0:
Ha := factor(product(1 + s/z,k = 0..N-1)/product(1 + s/p,k = 0..N))*eval(1/s∧alpha,s = wl):
return(Ha):
end proc:

Using α = 0.5, wl = 0.01, wh = 100 and ε = 2.36, we obtain N = 4 and the rational
approximation is computed as

alpha := 0.5 : wl := 0.01 : wh := 100 : epsilon := 2.36 :

Ha := Chare f _V1(alpha, wl, wh, epsilon);

Ha :=
13.3044(s + 0.0510506)(s + 0.448744)(s + 3.94457)(s + 34.6736)

(s + 0.0172187)(s + 0.151356)(s + 1.33045)(s + 11.6949)(s + 102.801)

Ha := expand(numer(Ha))/expand(denom(Ha));

Ha :=
13.3044s4 + 520.442s3 + 2076.76s2 + 921.238s + 41.6864

s5 + 115.995s4 + 1374.11s3 + 1828.18s2 + 273.172s + 4.16861

(25)

The magnitude and phase responses of (25) are again illustrated in Figure 2a,b, and
their errors are also shown in Figure 2b,d. All frequency responses are colored with a
brown line. Advantages of Charef’s Version 1 approximation method are that (1) for the
upper frequency limit wh, the magnitude error decreases, whereas for wl , the magnitude
behavior is similar to Oustaloup’s approximation, and (2) the number of iterations depends
on ε. Hence, the error increases or decreases in terms of α-dB. Disadvantages of the
approximation method are that (1) in the frequency band boundaries, the phase error
increases, as one can be seen in Figure 2d (brown line). In fact, the phase response at the
lower frequency limit wl has a similar behavior to Oustaloup’s method. (2) Additionally,
for the same number of iterations, the order of the rational approximation is slightly higher
than Oustaloup’s method, but slightly lower than the refined Oustaloup’s method.

3.4. Charef’s Approximation Version 2

For this method [6], the goal is to approximate the function

H(s) =
1

(1 + s
wl )

α
≈ ∏N−1

k=0

(
1 + s

z
)

∏N
k=0

(
1 + s

p

) , 0 < α < 1 (26)

Unlike (21), an arbitrary pole is here included to model the frequency behavior of the
transfer function. Note that in general, systems usually exhibits finite magnitude at very
low frequency. To compute the zeros and poles, (22)–(24) are also used. Therefore, based
on Table 8, (26) is coded in Maple 18 as depicted in Table 9.

Table 9. Code in Maple 18 environment of (26).

restart:Digits := 6:
Charef_V2 := proc(alpha,wl,wh,epsilon)
local p0,p,z,a,b,k,Ha,N;
p0 := wl*10∧(epsilon/(20*alpha)):
a := 10∧(epsilon/(10*(1-alpha))):
b := 10∧(epsilon/(10*alpha)):
N := ceil(log10(wh/p0)/log10(a*b)):
p := (a*b)∧k*p0: z := (a*b)∧k*a*p0:
Ha := factor(product(1 + s/z,k = 0..N − 1)/product(1 + s/p,k = 0..N)):
return(Ha):
end proc:

Using α = 0.5, wl = 0.01, wh = 100 and ε = 2.36, we obtain N = 4 and the rational
approximation is computed as

alpha := 0.5 : wl := 0.01 : wh := 100 : epsilon := 2.36 :

Ha := Chare f _V2(alpha, wl, wh, epsilon);

Ha :=
1.33044(s + 0.0510506)(s + 0.448744)(s + 3.94457)(s + 34.6736)

(s + 0.0172187)(s + 0.151356)(s + 1.33045)(s + 11.6949)(s + 102.801)

Ha := expand(numer(Ha))/expand(denom(Ha));

Ha :=
1.33044s4 + 52.0442s3 + 207.676s2 + 92.1238s + 4.16864

s5 + 115.995s4 + 1374.11s3 + 1828.18s2 + 273.172s + 4.16861

(27)

According to Figure 2 (maroon line), one can observe that for wh the frequency
behavior of the magnitude and phase error responses is equal to Charef’s Version 1 ap-
proximation, whereas for wl , the magnitude and phase errors have better behavior in
comparison with Charef’s Version 1 method. Another advantage of this method is that the
number of iterations depends on ε and, as a consequence, the error is controlled in terms
of α-dB. The disadvantages of the approximation method are the same as those described
in Charef’s Version 1 approximation method. Comparing (25) and (27), it can seen that the
only difference is the gain of the transfer functions.

3.5. Carlson’s Approximation

The method proposed in [7] was derived from a regular Newton process used for
iterative approximation of the α-th root. The starting point of the method is defined as

H(s) = G(s)α (28)

Defining the initial value as H0(s) = 1, the approximated rational function is ob-
tained as

Hk(s) = Hk−1(s)
(1− α)Hk−1(s)

1
α + (1 + α)G(s)

(1 + α)Hk−1(s)
1
α + (1− α)G(s)

∀ k = 1 . . . N (29)

where G(s) = s defines the derivative and G(s) = 1
s defines the integral. The code in Maple

18 environment of (29) is written in Table 10.

Table 10. Code in Maple 18 environment of (29).

restart:Digits := 6:
Carlson := proc(alpha,G,N)
local k,Ha := 1:
for k from 1 to N do

Ha := Ha*((1-alpha)*Ha∧(1/alpha) + (1+alpha)*G)/((1 + alpha)*Ha∧(1/alpha) + (1-alpha)*G):
od:
return(simplify(expand(numer(Ha))/expand(denom(Ha)))):
end proc:

For this case, we use α = 1
2 , G(s) = 1

s and N = 2. The rational approximation is
obtained as

alpha := 1/2 : G := 1/s : N := 2 :

Ha := Carlson(alpha, G, N);

Ha :=
s4 + 36s3 + 126s2 + 84s + 9

9s4 + 84s3 + 126s2 + 36s + 1
PrintSystem(ZeroPoleGain(eval f (Ha)));

Ha :=
0.111111(s + 32.1634)(s + 3)(s + 0.704088)(s + 0.132474)
(s + 7.54863)(s + 1.42028)(s + 0.333333)(s + 0.0310912)

(30)

the fourth line in (30) gets the zeros, poles and gain of the function described in the
third line and hence, the approximated function is built in terms of poles and zeros, as
described in the fifth line. The magnitude and phase responses of the arbitrary-order
integral approximation are depicted in Figure 2a,c and the error for each case is shown in
Figure 2b,d. Here, a gold color line is used to plot the waveform at each case. The main
advantage of this approximation method is that the magnitude and phase errors are almost
zero within a narrow bandwidth. However, if one wants to enlarge the bandwidth, it is
necessary to increase N. However, if N = 3, the order of the generated polynomials is 13 and
for N = 4 the order increases to 40, which is infeasible from the point of view of electronic
synthesis. Other disadvantages found are that (1) there is no control over the frequency

interval of interest (wl , wh), and (2) for some values of 0 < α ≤ 1 with one or two digits,
the method generates terms of s raised to a rational exponent. For instance, for α = 9

10
one obtains

Ha :=

(
s + 19

19s + 1

)1/9(
s3 + 38s2 + 361s

)
+ 361s2 + 6878s + 361(

s + 19
19s + 1

)1/9
(361s3 + 6878s2 + 361s) + 361s2 + 38s + 1

(31)

which becomes a difficult or impossible task during the electronic synthesis process. One
possible solution is to compute several rational transfer functions with α different [8]. Thus,
we can generate 1

s9/10 = 1
s5/10

1
s1/5

1
s1/5 , obtaining

Ha :=
(

s4 + 36s3 + 126s2 + 84s + 9
9s4 + 84s3 + 126s2 + 36s + 1

)
∗(

128s7 + 2610s6 + 11367s5 + 22410s4 + 23760s3 + 13752s2 + 3810s + 288
288s7 + 3810s6 + 13752s5 + 23760s4 + 22410s3 + 11367s2 + 2610s + 128

)2

(32)

(3) If α→0, then the order of the generated polynomials increases exponentially, as
one can verify in the second term of (32).

3.6. Matsuda’s Approximation

This approximation method was suggested in [9], which fits the original function
in a set of logarithmically spaced frequencies given by w = [w1, w2, w3 . . . wN+1] and
wa = [|wα

1 |, |wα
2 |, |wα

3 | . . . |wα
N+1|] and by using the following functions

d1,1 = |wα
1 | d1,2 = |wα

2 | d1,3 = |wα
3 | . . . d1,c = |wα

c | ∀ c = 1 . . . N + 1

d2,2 = w2−w1
d1,2−d1,1

d2,3 = w3−w1
d1,3−d1,1

...

d3,3 = w3−w2
d2,3−d2,2

dr,c =
wc−wr−1

dr−1,c−dr−1,r−1
∀ r = 2 . . . c

(33)

getting a superior triangular matrix as

D =


d1,1 d1,2 d1,3 . . . d1,N+1

0 d2,2 d2,3 . . . d2,N+1
0 0 d3,3 . . . d3,N+1
...

...
...

. . .
...

0 0 0 . . . dN+1,N+1

 (34)

where each element of the main diagonal of D is used to recursively approximate the
desired function as a continued fraction

Ha(s) = d1,1 +
s− w1

d2,2+

s− w2

d3,3+

s− w3

d4,4+
. . .

s− wN
dN+1,N+1

(35)

All equations mentioned before are coded in Maple 18 as given in Table 11.

Table 11. Matsuda’s method coded in Maple 18 environment.

restart:Digits := 6:
logspace := proc(a,b,n) evalf(10∧ ∼ <seq(evalf(a)..evalf(b),evalf((b − a)/(n − 1),7))>) end proc:
Matsuda := proc(alpha,wl,wh,N,k)
local w,wa,d,c,r,Ha,Dp := []:
w := convert(logspace(log10(wl),log10(wh),N + 1),list):
wa := [seq(abs(w[r]∧alpha),r = 1..nops(w))]:
d := matrix(nops(w),nops(w),0):
for c from 1 to N + 1 do

d[1,c] := wa[c]:
for r from 2 to c do

d[r,c] := (w[c] − w[r − 1])/(d[r − 1,c] − d[r − 1,r − 1]):
od:
Dp := [op(Dp),d[c,c]]:

od:
Ha := op(nops(Dp),Dp):
for r from nops(Dp) − 1 by −1 to 1 do

Ha := Dp[r] + (s-w[r])/Ha:
od:return(simplify(k*Ha)):
end proc:

The second line of Table 11 defines the logspace function. As an example on the use
of Table 11, we define α = −0.5, wl = 0.01, wh = 100, N = 8 and gain k = 1. Therefore, the
rational approximation is computed as

alpha := −0.5 : wl := 0.01 : wh := 100 : N := 8 : k := 1 :

Ha := Matsuda(alpha, wl, wh, N, k);

Ha :=
0.456s4 + 84.07s3 + 607.77s2 + 317.944s + 10.3411

10.3406s4 + 317.56s3 + 608.078s2 + 84.1832s + 0.45518
PrintSystem(ZeroPoleGain(eval f (Ha)));

Ha :=
0.111111(s + 32.1634)(s + 3)(s + 0.704088)(s + 0.132474)
(s + 7.54863)(s + 1.42028)(s + 0.333333)(s + 0.0310912)

(36)

The magnitude and phase responses of the approximation are depicted in Figure 3a,c,
and the error for each of them is also shown in Figure 3b,d. All frequency responses
are colored with a blue line. The main advantage of this approximation method is that
the magnitude and phase errors are low within the desired bandwidth and still can be
minimized by increasing the number of iterations N. In fact, the order of the rational
transfer function is always N/2. For this reason, N must be even. Otherwise, improper
transfer functions are generated. After several numerical simulations, we recommend
using, as a starting point, N = 8 in order to obtain magnitude and phase errors within
±2 dB and ±6◦, respectively. On the other hand, the main disadvantages found are that
(1) if N is very large, then high-order rational transfer functions are generated. This fact
becomes a tedious task during the electronic synthesis process. (2) Additionally, for the
parameters used, we note that from N = 16, some positive poles are generated, hence
conditioning the stability of the rational approximation.

(a) (b)

(c) (d)
Figure 3. Ideal behavior (black line), Matsuda’s (blue line), continued fraction expansion (red line),
curve fitting (brown line) and MSBL (maroon line) approximation for 1

s0.5 . (a) Magnitude response.
(b) Magnitude error. (c) Phase response. (d) Phase error.

3.7. Continued Fraction Expansion Approximation

This method is widely used for evaluation of functions since it converges faster than
power series expansions within a larger domain in the complex plane [28]. Continued
fraction expansion is defined as

(1 + x)α =
1

1−
αx
1+

(1 + α)x
2+

(1− α)x
3+

(2 + α)x
2+

(2− α)x
5+

. . .
(N + α)x

2+
(N − α)x
2N + 1

(37)

Replacing x = s− 1, we can approximate sα. To obtain the arbitrary-order integral,
i.e., 1

sα , the numerical value of α must be negative. Table 12 shows the code in Maple 8
of (37).

Table 12. Code in Maple 18 environment of (37).

restart:Digits := 6:with(DynamicSystems):with(numtheory):
CFE := proc(alpha,N)
local x,Ha:unassign(‘x’,‘s’):
Ha := cfrac(cfrac((x + 1)∧alpha,x,N)):x := s − 1:
return(expand(numer(eval(Ha)))/expand(denom(eval(Ha)))):
end proc:

Now, by defining α = −0.5 and N = 8, the rational transfer function is computed as

alpha := −0.5 : N := 8 :

Ha := CFE(alpha, N);

Ha :=
s4 + 36s3 + 126s2 + 84s + 9

9s4 + 84s3 + 126s2 + 36s + 1
PrintSystem(ZeroPoleGain(eval f (Ha)));

Ha :=
0.111111(s + 32.1634)(s + 3)(s + 0.704088)(s + 0.132474)
(s + 7.54863)(s + 1.42028)(s + 0.333333)(s + 0.0310912)

(38)

Note that for these parameters, (38) is equal to (30), but this is not always the case.
The magnitude and phase responses in the frequency domain of the rational transfer
function are depicted in Figure 3a,c and the error for each case is shown in Figure 3b,d,
respectively. In this case, a red line is used to plot all waveforms. After several numerical
simulations, one concluded that the main advantages of this approximation method are
that (1) the magnitude and phase errors are almost zero within a narrow bandwidth and
(2) the order = N/2 of the transfer function holds, where N is always even. In order to
extend the bandwidth, N must be increased. Similarly to Matsuda’s method, we also
recommend use, as starting point, N = 8 and hence, the magnitude and phase errors are
±1.5 and ±5◦, approximately. Moreover, the disadvantages found are that (1) there is no
control in the frequency range of interest (wl , wh), (2) if N is odd, then improper transfer
functions are generated, and (3) high-order rational polynomials are generated when N
becomes very large and consequently affects the electronic synthesis process.

3.8. Curve-Fitting Approximation

This approximation method was reported in [29,30], which is based on obtaining the
frequency response data of sα into a set of logarithmically spaced frequencies given as
w = [wl , w2, w3 . . . wh] ∈ (wl , wh). Next, by iteratively applying the Sanathanan–Koerner
least-squares method [31], the integer-order transfer function is approximated from the
frequencies response data and defined as

Ha(s) =
P(s)
Q(s)

≈ ∑N
n=0 pnsn

1 + ∑N
n=1 qnsn

≈ Pψ(s)
1 + Qφ(s)

(39)

where P, Q, ψ(s) and φ(s) are defined as

P = [p0, p1, . . . , pN ]
T Q = [q0, q1, . . . , qN ]

T

ψ(s) = [1, s, . . . , sN ] φ(s) = [s, s2, . . . , sN ]
(40)

The target is to obtain each coefficient of P and Q into the desired frequency range w.
To do, Levy’s linearized cost function along with the Sanathanan–Koerner least-squares
iteration are used to minimize the difference between data samples sα = (jwk)

α and (39) as

arg min
P,Q

h

∑
k=l

∣∣∣∣ P(jwk)

Qτ−1(jwk)
− Qτ(jwk)

Qτ−1(jwk)
(jwk)

α

∣∣∣∣2 (41)

where τ = 1 . . .> is the iteration step and > is the number of iterations. The curve fitting is
achieved when Qτ−1(jwk) approaches to Qτ(jwk), and for obtaining a minimal realization
of (39), equivalent poles and zeros must be canceled. The last step is to convert the obtained
state-space representation to a transfer function. Unfortunately, this method is complex
to code in Maple 18, but easy to implement in the Matlab environment, due to the use
of frd and fitfrd functions. However, it is possible to link Maple 18 with Matlab. The first
block of Table 13 shows the Maple 18 code to link and call the Matlab script shown in the
second block.



Fractal Fract. 2021, 5, 267 16 of 23

Table 13. Code in Maple 18 environment to call the m-file and Matlab script.

CF := proc(alpha,wl,wh,N)
Matlab[setvar](“alpha”,alpha):Matlab[setvar](“wl”,wl):
Matlab[setvar](“wh”,wh):Matlab[setvar](“N”,N):
Matlab[evalM](“CF(alpha,wl,wh,N)”);
end proc:

Matlab script of (39)–(41)

function Ha = CF(alpha,wl,wh,N)
w = logspace(log10(wl),log10(wh));
A = (j*w).∧alpha;
Ha = fitfrd(frd(A,w),N);
[num,den] = ss2tf(Ha.A,Ha.B,Ha.C,Ha.D);
Ha = minreal(tf(num,den))
end

We now define α = −0.5, wl = 0.01, wh = 100 and N = 4. The rational approximation is
computed as

alpha := −0.5 : wl := 0.01 : wh := 100 : N := 4 :

Ha := CF(alpha, wl, wh, N);

Ha :=
0.08931s4 + 4.331s3 + 10.74s2 + 2.692s + 0.05927

s4 + 9.209s3 + 7.152s2 + 0.5605s + 0.002899
PrintSystem(ZeroPoleGain(eval f (Ha)));

Ha :=
0.0893100(s + 45.8877)(s + 2.32684)(s + 0.255132)(s + 0.0243616)

(s + 8.36168)(s + 0.759744)(s + 0.0820100)(s + 0.00556443)

(42)

The magnitude and phase responses are depicted in Figure 3a,c, and the errors are
shown in Figure 3b,d. All the curves are colored with a brown line. The main advantages
of this approximation method are that (1) if N is low, for instance N = 4, the magnitude
error is low into the desired frequency range, ±2 dB approximately, and the ripples can
be minimized by increasing N. (2) If N is large, the phase error becomes very low into the
frequency interval. The disadvantages found are that (1) if N is low, for instance N = 4 ,
then close to the wh upper limit, the ripple of the phase error is high, −25◦ approximately.
(2) Additionally, when N increases, the magnitude error at wh is slightly higher than the
error at wl . Note that if N becomes very large, then the electronic synthesis process could
be a tedious and complex task.

3.9. Modified Stability Boundary Locus (MSBL) Fitting Approximation

This approximation method was reported in [19,32], which is an improved version of the
stability boundaries locus fitting approximation method [33]. According to [19], the method uses
a set of logarithmically spaced frequency sampling points wk = [wl , w2, w3 . . . wh] ∈ (wl , wh),
for approximating the behavior of sα. A closed-loop control system composed by the plant,
i.e., sα, and a proportional–integral controller is assumed. Following the analysis of [19,33],
the equations of the method are

Ar,k = (jwk)
r+1 − (jwk)

N−r+1 cos
(

π
2 α
)

wα
k
− (jwk)

N−r sin
(

π
2 α
)

wα−1
k

= Re(Ar,k) + Im(Ar,k)

Bk = −(jwk)
N+1 + jwk

cos
(

π
2 α
)

wα
k

+
sin
(

π
2 α
)

wα−1
k

= Re(Bk) + Im(Bk)

(43)

and the unknown coefficients of the approximate model are computed as

C = [a0, a1, . . . , aN−1] =

[
1

[Ar,k]T
· [Bk]

T
]T

(44)

and the approximated integer-order rational transfer function is given by

Ha(s) =
a0sN + a1sN−1 + . . . + aN−1s + 1
sN + aN−1sN−1 + . . . + a1s + a0

(45)

Equations (43)–(45) are coded in Maple 18 as shown in Table 14.

Table 14. Maple 18 code of MSBL method.

restart:Digits := 6:with(LinearAlgebra):
logspace := proc(a,b,n) evalf(10∧ ∼ <seq(evalf(a)..evalf(b),evalf((b-a)/(n-1),7))>) end proc:
MSBL := proc(alpha,wl,wh,N)
local A := Matrix(N),B := Vector[row](N),C,k,r,w,Ha:unassign(‘s’):
if N = 1 then w[1] := wh:
else w := convert(logspace(log10(wl),log10(wh),N),list): fi:
for k from 1 to N do

for r from 1 to N do
A[r,k] := (I*w[k])∧r-(I*w[k])∧(N-r + 2)*cos(Pi/2*alpha)/w[k]∧alpha

–(I*w[k])∧(N-r + 1)*sin(Pi/2*alpha)/w[k]∧(alpha-1):
A[r,k] := Re(A[r,k]) + Im(A[r,k]):
B[k] := –(I*w[k])∧(N + 1) + I*w[k]*cos(Pi/2*alpha)/w[k]∧alpha

+sin(Pi/2*alpha)/w[k]∧(alpha-1):
B[k] := Re(B[k]) + Im(B[k]):

od:
od:unassign(’k’):
C := convert(((1/A)∧%T.B∧%T)∧%T,list):
Ha := sort(sum(1/N + C[k + 1]*s∧(N-k),k = 0..N-1),s,descending)/

sort(sum(s∧N/N + C[N-k]*s∧(N-k-1),k = 0..N-1),s,descending):
return(Ha):
end proc:

By using α = −0.5, wl = 0.01, wh = 100 and N = 4, the rational approximation is
computed as

alpha := −0.5 : wl := 0.01 : wh := 100 : N := 4 :

Ha := MSBL(alpha, wl, wh, N);

Ha :=
0.05266 ∗ s4 + 9.46702s3 + 68.5199s2 + 34.5218s + 1

s4 + 34.5218s3 + 68.5199s2 + 9.46702s + 0.05266
PrintSystem(ZeroPoleGain(eval f (Ha)));

Ha :=
0.0526682(s + 172.216)(s + 6.99006)(s + 0.511296)(s + 0.0308478)

(s + 32.4172)(s + 1.95581)(s + 0.143060)(s + 0.00580665)

(46)

The magnitude and phase responses are illustrated in Figure 3a,c and the errors are
depicted in Figure 3b,d, respectively. For this last method, all curves are colored with a
maroon line. The main advantages of this approximation method are that (1) the magnitude
and phase responses are improving when N increases and for the parameters described
before, the magnitude error is ±1.5 dB and for the phase error is ±7◦, approximately, and
(2) the method can not only be used to approximate arbitrary-order operators, but can also
be used to approximate transfer functions. The main disadvantage found is that wl should
be set close to zero since if wl drifts toward higher frequencies, then the method fails.

It is worth mentioning that other approximation methods of rational transfer functions can
be found in the literature, such as the Fourier series and inverse Fourier transform method [20],

vector fitting method [34], Abdelbaki’s method [35], Maione’s approximation [36], Thiele’s
approximation [37], El-Khazali’s method [38], AbdelAty’s method [39]. However, the
accuracy of each method is almost the same or less with some method described above.
For this reason, these approximation methods are not discussed herein.

4. Model Order Reduction

From the point of view of electronic design, low-order accurate transfer functions
are widely required. This is because they are easier to synthesize with analog circuits or
be encoded with reconfigurable hardware. In this context, two model order reduction
methods are described below.

4.1. Pade’s Approximation

The Pade’s method approximates a continuous function by a rational function of
[m/n] order [40]. Given a function Ha(s) with two integers 0 ≤ m, n and n ≤ m, Pade’s
approximation is defined as

P(s) =
∑m

j=0 ajsj

1 + ∑n
k=1 bksk =

a0 + a1s + a2s2 + . . . + amsm

1 + b1s + b2s2 + . . . + bnsn (47)

and have as many of their derivatives as possible equal to s = 0.

Ha(0) = P(0)

H′′a (0) = P′′(0)
...

Hm+n
a (0) = Pm+n(0)

(48)

Note that for k = 0, the obtained approximation is the Maclaurin expansion for Ha(s).
In the Maple 18 environment, Pade’s approximation can be computed as given in Table 15.

Table 15. Maple 18 code of Pade’s method.

restart:Digits := 6:with(SignalProcessing):
Pade := proc(Ha,m,n)
local A,Hp:unassign(‘s’):
A := series(Ha,s = 0.200):
Hp := convert(A,ratpoly,m,n):
end proc:

Applying the algorithm of Table 15 to each rational transfer function associated with
each approximation method, we obtain

POus(s) =
0.100001 + 6.23548s + 34.6971s2 + 15.7445s3

1 + 19.6351s + 33.0972s2 + 2.83636s3

PRe f _Ous(s) =
21.0820s + 464.965s2 − 91901.8s3 − 3.53478E6s4 − 3.52981E7s5

1 + 201.551s− 7544.47s2 − 717282s3 − 1.39038E7s4 − 5.7186E7s5

PChare f _V1(s) =
10.0001 + 425.430s + 4720.84s2 + 4514.38s3

0.999999 + 85.9742s + 1748.72s2 + 7433.33s3 − 1830.34s4

PChare f _V2(s) =
1.00001 + 42.5430s + 472.084s2 + 451.438s3

0.999999 + 85.9742s + 1748.72s2 + 7433.33s3 − 1830.34s4

PCarlson(s) =
8.99996 + 68.5714s− 5.62857s2 − 72.8571s3

0.999999 + 34.2857s + 65.6631s2 − 83.4526s3

(49)

PMatsuda(s) =
22.7187 + 529.917s− 4982.07s2 − 35993.4s3

0.999997 + 177.524s− 86.4722s2 − 18084.8s3

PCFE(s) =
8.99996 + 68.5714s− 5.62857s2 − 72.8571s3

0.999999 + 34.2857s + 65.6631s2 − 83.4526s3

PCF(s) =
20.4450 + 455.702s− 16273.1s2 − 29078.0s3

1 + 170.213s− 1931.47s2 − 40335.3s3

PMSBL(s) =
18.9868 + 467.349s− 4638.09s2 + 2698.41s3

0.999999 + 169.841s− 450.533s2 − 7149.58s3

(50)

As one can observe, from each original rational transfer function, Pade’s method
is applied to reduce only an order in each polynomial. According to Figure 4a,b, the
magnitude bandwidth is not only reduced, but for some approximation methods, the
magnitude error is increased drastically in the desired frequency range (wl , wh). This same
behavior is observed in the phase responses, as illustrated in Figure 4c,d. Of all of them, the
Oustaloup’s, refined Oustaloup’s and Charef’s V2 methods have the least errors one decade
down. Therefore, we conclude that Pade’s method fails because there is no approximation
control in the frequency interval.

(a) Magnitude responses of (49) (b) Magnitude responses of (50)

(c) Phase responses of (49) (d) Phase responses of (50)
Figure 4. Magnitude and phase responses of reduced order transfer functions by applying Pade’s
approximation. (a) Ideal behavior (black line) of 1

s0.5 , Oustaloup’s (blue line), refined Oustaloup’s
(red line), Charef’s V1 (brown line), ideal behavior (gray line) of 1

1+s0.5 , Charef’s V2 (maroon line) and
Carlson’s (gold line), (b) ideal behavior (black line) of 1

s0.5 , Matsuda’s (blue line), continued fraction
expansion (red line), curve fitting (brown line) and MSBL (maroon line), (c) phase responses of (a,d),
phase responses of (b).

4.2. Stochastic Balancing Method

The balred function of Matlab defines the stochastic balancing related model reduction
method [41,42]. Applying this function to each rational transfer function associated with
each approximation method as

Pr = balred(Ha, R,′ FreqIntervals′, [wl wh]) (51)

where Ha is the original transfer function, Pr is the reduced order transfer function and R
is the order, we obtain

PrOus(s) =
9.827s3 + 151.5s2 + 143.8s + 4.139

s3 + 55.41s2 + 213.2s + 41.39

PrRe f _Ous(s) =
18.77s5 + 2142s4 + 33580s3 + 49910s2 + 5988s

s5 + 2612s4 + 11780s3 + 58880s2 + 25540s + 640.1

PrChare f _V1(s) =
0.1826s3 + 2.012s2 + 1.052s + 0.04883

s3 + 1.99s2 + .3172s + 0.004883

PrChare f _V2(s) =
0.01826s3 + 0.2012s2 + 0.1052s + 0.004883

s3 + 1.99s2 + 0.3172s + 0.004883

PrCarlson(s) =
0.1524s3 + 2.604s2 + 2.737s + 0.3217

s3 + 3.518s2 + 1.257s + 0.03574

(52)

PrMatsuda(s) =
0.1716s3 + 2.196s2 + 1.218s + 0.03979

s3 + 2.297s2 + 0.3237s + 0.001751

PrCFE(s) =
0.1524s3 + 2.604s2 + 2.737s + 0.3217

s3 + 3.518s2 + 1.257s + 0.03574

PrCF(s) =
0.2519s3 + 1.52s2 + 0.4432s + 0.009953

s3 + 1.129s2 + 0.09365s + 0.0004868

PrMSBL(s) =
0.1884s3 + 2.262s2 + 1.199s + 0.03487

s3 + 2.35s2 + 0.3298s + 0.001836

(53)

Similarly as before, the stochastic balancing method is applied to reduce an order
in each numerator and denominator of the original transfer function. Thus, whereas
the magnitude responses of (52) are depicted in Figure 5a, those of (53) are illustrated
in Figure 5b. At this stage, one can observe that for Oustaloup’s and refined Oustloup’s
methods, the bandwidth is slightly kept in the desired frequency range. However, for
phase responses, the bandwidth of all approximation methods is reduced, as shown in
Figure 5c,d.

(a) Magnitude responses of (52). (b) Magnitude responses of (53).

Figure 5. Cont.

(c) Phase responses of (52). (d) Phase responses of (53).

Figure 5. Magnitude and phase diagrams of reduced order transfer functions by applying stochastic
balancing method. (a) Ideal behavior (black line) of 1

s0.5 , Oustaloup’s (blue line), refined Oustaloup’s
(red line), Charef’s V1 (brown line), ideal behavior (gray line) of 1

1+s0.5 , Charef’s V2 (maroon line) and
Carlson’s (gold line), (b) ideal behavior (black line) of 1

s0.5 , Matsuda’s (blue line), continued fraction
expansion (red line), curve fitting (brown line) and MSBL (maroon line), (c) phase responses of (a,d),
phase responses of (b).

5. Results and Conclusions

Three basic definitions of arbitrary-order calculus were revised, and the Maple 18
code for each of them along with an application example were shown. Afterwards, the
frequency-domain approximation methods were introduced. For each approximation
method, not only the Maple 18 code was presented, but an application example was
also explained, where the closed form transfer functions were derived for α = 1/2. The
advantages and disadvantages of each approximation method were also discussed. From
the frequency response waveforms, it is evident that curve fitting and MSBL methods
are better suited to the ideal behavior. On the one hand, if the required bandwidth for a
specific application is narrow, then continued fraction expansion or Matsuda’s method
can be used. On the other hand, if the application needs to control the approximation
error, then Charef’s V1 and V2 methods should be used. Furthermore, we also note that
in general, the accuracy of the approximation methods depends on N (ε for Charef’s V1
and V2) and the desired frequency interval. Therefore, for a large bandwidth, N should
be increased to obtain the desired accuracy, but if the bandwidth is narrow, then N can
be reduced. For all approximation methods, the rational transfer function as a product of
poles and zeros was also derived. This is important because each pole–zero pair can be
synthesized in the analog domain by using first-order shelving equalizers, and for the case
of a pole with gain, a low-pass filter topology can be used. Finally, we showed two model
order reduction methods and according to the numerical results, the stochastic balancing
method achieves the best reduced-order transfer functions, despite all its disadvantages.
As a consequence, the electronic synthesis process will be accelerated.

